

Getting Started with Instrument Control Using Microsoft Visual Studio's VB.NET Language

Intro

Are you tasked with automating the setup and control of instruments for a test sequence? Do you prefer the Visual Basic (VB) programming language over others but not sure where to start? Whether you are a new programmer looking for some help and stepping stones, or a seasoned Test Engineer looking for just a hint of guidance, this piece may be just the thing for you. In the following pages you will find:

- A general overview of the downloading and installation of the Microsoft Visual Studio 2019 Integrated Development Environment
- Steps for starting up your first VB .NET Framework project
- Steps for adding the VISA COM reference tools
- An overview on the general building blocks you will need for connecting to, sending commands to, and receiving data from your connected instrumentation.
- Building and running your simple instrument-controlling VB application

For brevity, we have specifically skipped discussing things like good coding practices, syntax and how to do what with VB, and any debugging information. There are plenty of online resources for you to refer to without us recreating those discussion points here.

Prerequisites


Prior to progressing through this document, you will need to ensure you have some programming knowledge and additional necessary tools installed on your computer. These include:

- A general or basic background in computer programming
- National Instruments VISA 17.5 (or higher) Runtime or Full Version

Download Visual Studio Community

Download the free version of Visual Studio from the Microsoft website:

<https://visualstudio.microsoft.com/vs/community/>




What you will need to use the VS Community version:

- A pulse
- A Hotmail or Outlook account
- A passion for writing code
- A network connection

Install and Select the Basic Programming Tools


Run the installer and it will perform a quick check of your computer so that it can download the necessary installation items for the setup.




You will then be presented with the following screen which presents you with options for customizing your VS installation. Because we are only concerning ourselves with programming in VB.NET, only choose the following three items for now (outlined in green in the image):

- .NET desktop development
- Desktop development with C++
- Universal Windows Platform development


Note that we have specifically deterred you from including the Python development as part of the installation. While Python is a powerful language itself, we avoid adding per this installer so that any existing Python version does not face a secondary installation to contend with. We encourage you to explore this in the future but hold off for the purposes of this exercise.




After clicking the Install button, you will see the VS version begin its download, verify, and install of the software tools.



As with most successful Windows application installations, a reboot is required at some point.




To verify a successful installation, search for Visual Studio 2019 in your list of installed applications and launch.



Create Your First VB.NET Program

Create the starter project

After Visual Studio launches, you will locate and click the **Create a new project** button. From there you will need to locate the Console App (.NET Framework) option; select it; and click **Next**.



Next, name your project using the input field provided. To keep things consistent, enter "My_First_VB_Project" as the name then click the **Create** button.

Configure your new project

Console App (.NET Framework)

Visual Basic

Windows


Console

Project name

My_First_VB_Project

Location

C:\Users\██████\source\repos

Solution name 


My_First_VB_Project

☐ Place solution and project in the same directory

Framework

.NET Framework 4.7.2


At this point, your project will be created, and you will be presented with the basic setup for your Module1.vb file – the main source file for your VB program.




Note that while we see the traditional route is to do something with “Hello World!”, we will be skipping this and going much further.

Adding the VISA Reference

On the right-hand side of the Visual Studio interface you should see a pane entitled Solution Explorer. From the items displayed, right-click on the **References** text and navigate to **Add Reference**.




Select the COM category and scroll down the list of names to locate the **VISA COM 5.9 Type Library**; add a check to the box adjacent to it; and click the **OK** button.



Side note, but perhaps your VISA COM Type Library is not 5.9 as seen in this document. This is likely because you have a different VISA version installed than the one we listed above in the Prerequisites section. No matter – if you have the VISA COM Type Library of any version go ahead and add it as it is a safe assumption it will work. While there are likely some changes in the underlying code in the different versions you will most likely find that there is little or no difference in the tools implemented here.

Finally, to use the tools within the VISA COM library, you will need to tell your source file to use it as well. You will do this by entering the text “imports Ivi.Visa.Interop;” at the end of the list of other pre-populated reference items in your editor:



A Quick Note on the Use of Wrapper Functions

Appendix A is populated with a completed and more elaborate version of the example you are about to read through. One of the things you should take note of are the series of wrapper functions – functions which bundle other functions and variables – used in this example. These have been provided to help bring some additional clarification to the VISA tools they leverage. While the Visual Studio Intellisense feature shares brief pop-up information and suggestions as you type in your commands, the wrapper functions share more verbose information per the comments found within their declarations. Along with sharing details of the function purpose, parameters, return value, and revision details, using the wrappers enables us to customize the use of reference library methods and attributes while also providing additional information to users of all skill levels.

For instance, a common operation used with instrument control is a **query**. Think of this as asking a question: you make a statement in the form of a question and expect an answer. While some VISA API implementations have a query function built in, the VISA COM library used in this Visual Studio example does not. However, we can (and have) created a query function of our own using the VISA `WriteString()` and `ReadString()` functions:

```
1 reference
Function Instrument_Query(instrument_control_object As FormattedIO488, command As String) As String
|
```

We will leave the task of research and investigation of the VISA COM library options and the VB.NET programming language in your hands.

Build the Basic Control Tools with Our Template Functions

Basic control of your instrument can be broken down into three main elements:

- Connecting to your instrument
- Sending and receiving information (settings, readings, etc.)
- Disconnecting from your instrument

The following steps will guide you through each of these.

Start by adding a resource manager object just after the `Main()` declaration:

```

References
Sub Main()
    '
    ' First step: Open the resource manager And assigns it to an object variable
    Dim resource_manager As Ivi.Visa.Interop.ResourceManager = New Ivi.Visa.Interop.ResourceManager

```


Yes, this does seem like an extra step not listed in the three bullets above. However, the Resource Manager is the layer of VISA responsible for managing connections to instruments and identifying attributes about them in a relatively non-invasive manner. It can look at all resources connected to your computer without disrupting other programs which might be using them. You do, however, need it active prior to taking control of your target instrument.

With the Resource Manager activated – or instantiated – you can now create an object that will be used to connect to and manage the transactions with your instrument. Below you will notice that `my_instrument` is the object which is first created to become your instrument connection. The instrument identification string tells the Resource Manager what communications protocol is being used (GPIB, USB, Ethernet, or RS232) and what address at which to look for and identify the instrument you want to control. We use the Resource Manager and `FormattedIO488` (instrument) objects along with the instrument identification string to make the connection to the instrument.

```
,  
' Second step: Create a FormattedIO488 object to represent the instrument  
' you intend to communicate with, And connect to it.  
,  
  
Dim my_instrument As FormattedIO488 = New Ivi.Visa.Interop.FormattedIO488()  
Dim instrument_id_string As String = "USB0::0x05E6::0x6510::04340543::INSTR"  
Dim timeout As Int16 = 20000 ' define the timeout in terms of milliseconds  
' Instrument ID String examples...  
' LAN -> TCP/IP0:134.63.71.209:inst0 : InStr()  
' USB -> USB0:0x05E6:0x2450:01419962:InStr()  
' GPIB -> GPIB0:16:InStr()  
' Serial -> ASRL4:InStr()  
Connect_To_Instrument(resource_manager, my_instrument, instrument_id_string, timeout)
```

After you connect to the instrument you can build your test sequence with your choice of commands written to the equipment and in some cases reading information back from it. As noted earlier, a query is the combination of a write followed by a read. Below we loop a query for the instrument identification string to be reported back to the program so that it can be printed to the program console.

```
,  
' Third step: Issue commands to the instrument And receive responses to  
' be printed to the program console window.  
,  
  
For i As Int16 = 0 To 10 Step 1  
    Console.WriteLine(Instrument_Query(my_instrument, "*IDN?"))  
Next
```


After your test sequence is complete, it is good practice – and sometimes a necessity – to disconnect from the instrument.

```

'
' Fourth step: Close the instrument object And release it for use
' by other programs.
'
Disconnect_From_Instrument(my_instrument)

```

You can then build your program by navigating to **Build->Build Solution** or using the Ctrl+Shift+B hotkey combination.



The screenshot shows the Visual Studio Output window with the 'Build' tab selected. The output text is as follows:

```

1>----- Build started: Project: My_First_VB_Project, Configuration: Debug Any CPU -----
1> My_First_VB_Project -> C:\Users\██████████\source\repos\My_First_VB_Project\My_First_VB_Projec
===== Build: 1 succeeded, 0 failed, 0 up-to-date, 0 skipped =====

```

You can then run your program by navigating to **Debug->Start Debugging** or using the F5 hotkey. When you do, you will see that the instrument you have connected to reports its identification string.

```

*IDN?
KEITHLEY INSTRUMENTS,MODEL DAQ6510,04340543,1.0.04b

```

There is plenty more to learn and understand with respect to VB.NET programming and using the Visual Studio IDE (things like setting breakpoints, stepping through and into your code line-by-line, etc.), but we leave that for you to learn on your own. Your challenge – using the tools and information presented here – is to collect the series of commands you will need to control your equipment and successfully implement them. Good luck!

Conclusion

As noted earlier, we have only covered the very basic necessities of remote instrument programming here. If you have not done so already, we strongly encourage you to locate a VB.NET programmer's reference to have on hand as you continue your journey. Below (in the Appendix) you will find a completer and more elaborate version of the example we just reviewed from which you can copy and paste the core functions and their usage.

Appendix A – Example Code

The following code is the basic construction that this document walks the reader through. This code (along with numerous other examples) can be found at <https://github.com/tektronix/keithley>.

```

'=====
' Copyright 2019 Tektronix, Inc.
' See www.tek.com/sample-license for licensing terms.

```

```

'=====

Imports Ivi.Visa.Interop

Module Module1
    Private ReadOnly echo_command As Boolean = True

    Sub Main()
        ' Create a Stopwatch object And capture the program start time from the system.
        Dim myStpWtch As Stopwatch = New Stopwatch()
        myStpWtch.Start()

        '
        ' First step: Open the resource manager And assigns it to an object variable
        '
        Dim resource_manager As Ivi.Visa.Interop.ResourceManager = New Ivi.Visa.Interop.ResourceManager

        '
        ' Second step: Create a FormattedIO488 object to represent the instrument
        ' you intend to communicate with, And connect to it.
        '
        Dim my_instrument As FormattedIO488 = New Ivi.Visa.Interop.FormattedIO488()
        Dim instrument_id_string As String = "USB0::0x05E6::0x6510::04340543::INSTR"
        Dim timeout As Int16 = 20000 ' define the timeout in terms of milliseconds
        ' Instrument ID String examples...
        ' LAN -> TCPIP0:134.63.71.209:inst0 : InStr()
        ' USB -> USB0:0x05E6:0x2450:01419962:InStr()
        ' GPIB -> GPIB0:16:InStr()
        ' Serial -> ASRL4:InStr()
        Connect_To_Instrument(resource_manager, my_instrument, instrument_id_string, timeout)

        '
        ' Third step: Issue commands to the instrument And receive responses to
        ' be printed to the program console window.
        '
        For i As Int16 = 0 To 10 Step 1
            Console.WriteLine(Instrument_Query(my_instrument, "*IDN?"))
        Next

        '
        ' Fourth step: Close the instrument object And release it for use
        ' by other programs.
        '
        Disconnect_From_Instrument(my_instrument)

        ' Capture the program stop time from the system.
        myStpWtch.Stop()

        ' Get the elapsed time as a TimeSpan value.
        Dim ts As TimeSpan = myStpWtch.Elapsed

        ' Format And display the TimeSpan value.
        Dim elapsedTime As String = $"{ts.Days:00}:{ts.Hours:00}:{ts.Minutes:00}:{ts.Seconds:00}.{ts.Milliseconds _
            / 10:000}"
        Console.WriteLine("RunTime " + elapsedTime)

        ' Implement a keypress capture so that the user can see the output of their program.
        Console.WriteLine("Press any key to continue...")
        Dim k As Char = Console.ReadKey().KeyChar
    End Sub

    Sub Connect_To_Instrument(ByRef resource_manager As ResourceManager, ByRef instrument_control_object As
FormattedIO488, instrument_id_string As String, timeout As Int16)
        '
        ' Purpose: Open an instance Of an instrument Object For remote communication And establish the communication
        attributes.
        '
        ' Parameters:
        ' resource_manager - The reference to the resource manager object created external to this function. It Is
        passed in
        ' by reference so that any internal attributes that are updated when using to connect to
        the

```

```

'
'           instrument are updated to the caller.
'
'           instrument_control_object - The reference to the instrument object created external to this function. It
Is passed
'           in by reference so that it retains all values upon exiting this function,
making it
'           consumable to all other calling functions.
'
'           instrument_id_string - The instrument VISA resource string used to identify the equipment at the
underlying driver
'           level. This string can be obtained per making a call to Find_Resources() VISA
function And
'           extracted from the reported list.
'
'           timeout - This Is used to define the duration of wait time that will transpire with respect to VISA
read/query calls
'           prior to an error being reported.
'
' Returns:
'     None
'
' Revisions:
'     2019-06-14      JJB      Initial revision.
'
instrument_control_object.IO = resource_manager.Open(instrument_id_string, Ivi.Visa.Interop.AccessMode.NO_LOCK,
20000)
' Instrument ID String examples...
'     LAN -> TCP/IP0:134.63.71.209:inst0:INSTR
'     USB -> USB0:0x05E6:0x2450:01419962:INSTR
'     GPIB -> GPIB0:16:INSTR
'     Serial -> ASRL4:INSTR
instrument_control_object.IO.Clear()
Dim myTO As Int16 = instrument_control_object.IO.Timeout
instrument_control_object.IO.Timeout = timeout
myTO = instrument_control_object.IO.Timeout
instrument_control_object.IO.TerminationCharacterEnabled = True
instrument_control_object.IO.TerminationCharacter = Convert.ToByte(10)
Return
End Sub

Sub Disconnect_From_Instrument(ByRef instrument_control_object As FormattedIO488)
'
' Purpose: Closes an instance Of And instrument Object previously opened For remote communication.
'
' Parameters:
'     instrument_control_object - The reference to the instrument object created external to this function. It
Is passed
'     in by reference so that it retains all values upon exiting this function,
making it
'     consumable to all other calling functions.
'
' Returns:
'     None
'
' Revisions:
'     2019-06-14      JJB      Initial revision.
'
instrument_control_object.IO.Close()
Return
End Sub

Sub Instrument_Write(instrument_control_object As FormattedIO488, command As String)
'
' Purpose: Used to send commands to the instrument.
'
' Parameters:
'     instrument_control_object - The reference to the instrument object created external to this function. It
Is passed
'     in by reference so that it retains all values upon exiting this function,
making it
'     consumable to all other calling functions.
'
'     command - The command string issued to the instrument in order to perform an action.

```

```

'
' Returns
'     None
'
' Revisions
'     2019-06-04      JJB      Initial revision.
'
If (echo_command = True) Then
    Console.WriteLine("{0}", command)
End If
instrument_control_object.WriteString(command)
Return
End Sub

Function Instrument_Read(instrument_control_object As FormattedIO488) As String
'
' Purpose: Used to read commands from the instrument.
'
' Parameters:
'     instrument_control_object - The reference to the instrument object created external to this function. It
Is passed
'                                     in by reference so that it retains all values upon exiting this function,
making it
'                                     consumable to all other calling functions.
'
' Returns:
'     The string obtained from the instrument.
'
' Revisions:
'     2019-06-04      JJB      Initial revision.
'
Return instrument_control_object.ReadString()
End Function

Function Instrument_Query(instrument_control_object As FormattedIO488, command As String) As String
'
' Purpose: Used to send commands to the instrument And obtain an information string from the instrument.
'     Note that the information received will depend on the command sent And will be in string
'     format.
'
' Parameters:
'     instrument_control_object - The reference to the instrument object created external to this function. It
Is passed
'                                     in by reference so that it retains all values upon exiting this function,
making it
'                                     consumable to all other calling functions.
'
'     command - The command string issued to the instrument in order to perform an action.
'
' Returns:
'     The string obtained from the instrument.
'
' Revisions:
'     2019-06-04      JJB      Initial revision.
'
Instrument_Write(instrument_control_object, command)
Return Instrument_Read(instrument_control_object)
End Function
End Module

```

References

TBD

Revisions

Revision	Date	Authored by	Notes
1.0	2019-06-14	Josh Brown	Hey, you gotta start somewhere